
Pfad Accounting & Controlling
Schweizer Schreibweise: Semikolon, Punkt als Dezimalzeichen, Hochkomma als Tausender

10 Themen in der Reihenfolge des Lernpfads. Formel beginnt mit =, Argumente mit ; trennen, Text in
"Anführungszeichen".

[ ] = optionales Argument

01 Budget-Ist-Abweichung
Abweichung CHF: =Ist-Budget
Abweichung %: =(Ist-Budget)/Budget
BEISPIEL B2 = 12'000 (Budget), C2 = 13'450 (Ist):
=C2-B2 / =(C2-B2)/B2 → 1'450.00 / 0.1208 = 12.1 %
TYPISCHER FEHLER Zusätzlich mal 100 rechnen und
Prozentformat setzen ergibt 1'208 %.

02 Anteil am Total
=Wert/$Total$ (Total absolut fixiert)
BEISPIEL B2 = 4'250'000, B6 = 10'000'000 (Total): =B2/$B$6
→ 0.425 = 42.5 %
TYPISCHER FEHLER Ohne $ rutscht das Total beim Kopieren
nach unten mit und die Formel liefert #DIV/0!.

03 MwSt netto/brutto
Brutto: =Netto*(1+Satz)
Netto: =Brutto/(1+Satz) · MwSt: =Brutto-Brutto/(1+Satz)
BEISPIEL B2 = 1'081.00 (brutto), F1 = 0.081: =B2/(1+$F$1) →
1'000.00
TYPISCHER FEHLER Brutto*(1-Satz) ergibt nicht den
Nettobetrag (hier 993.44 statt 1'000.00).

04 Rundung 5 Rappen
=RUNDEN(Zahl/0.05;0)*0.05
oder =VRUNDEN(Zahl;Vielfaches)
BEISPIEL B2 = 47.33: =RUNDEN(B2/0.05;0)*0.05 → 47.35
TYPISCHER FEHLER =RUNDEN(B2;2) rundet nur auf Rappen,
nicht auf 5 Rappen.

05 SUMMEWENNS
=SUMMEWENNS(Summe_Bereich;Kriterien_Bereich1;
Kriterien1;
[Kriterien_Bereich2;Kriterien2];…)
BEISPIEL A = Kostenstelle, B = Monat, C = Betrag:
=SUMMEWENNS(C2:C9;A2:A9;"Marketing";B2:B9;"Feb")
→ 4'730.00
TYPISCHER FEHLER Der Summenbereich kommt hier zuerst,
anders als bei SUMMEWENN.

06 Duplikate finden
=WENN(ZÄHLENWENN($Bereich$;Zelle)>1;"Duplikat";
"")
BEISPIEL A2:A7 = Belegnummern, B-1041 kommt zweimal vor:
=WENN(ZÄHLENWENN($A$2:$A$7;A2)>1;"Duplikat";"") →
Duplikat
TYPISCHER FEHLER Ohne $ verschiebt sich der Prüfbereich
beim Kopieren und Duplikate weiter oben fallen heraus.

07 Plausibilitätscheck
=WENN(RUNDEN(SUMME(Soll);2)=RUNDEN(SUMME(Haben);
2);"OK";"Differenz")
BEISPIEL Soll B2:B6 und Haben C2:C6 je 22'350.75:
=WENN(RUNDEN(SUMME(B2:B6);2)=RUNDEN(SUMME(C2:C6);
2);"OK";"Differenz") → OK
TYPISCHER FEHLER Ohne RUNDEN melden Gleitkomma-Reste
(0.0000000001) eine Differenz, obwohl die Beträge stimmen.

08 Kontenabstimmung
=WENNFEHLER(SVERWEIS(Referenz;$Liste$;2;FALSCH);
"fehlt")
BEISPIEL A2 = ZA-7781, Hauptbuch E2:F7 ohne ZA-7781:
=WENNFEHLER(SVERWEIS(A2;$E$2:$F$7;2;FALSCH);
"fehlt") → fehlt
TYPISCHER FEHLER Ohne FALSCH findet SVERWEIS den
nächstkleineren Eintrag und die fehlende Buchung bleibt
unentdeckt.

09 Alterung offener Posten
=WENN(Stichtag-Fällig<=30;"0-30";
WENN(Stichtag-Fällig<=60;"31-60";">60"))
BEISPIEL H1 = 31.10.2026 (Stichtag), B2 = 12.09.2026 → 49
Tage: =WENN($H$1-B2<=30;"0-30";WENN($H$1-B2<=60;
"31-60";">60")) → 31-60
TYPISCHER FEHLER HEUTE() statt fixem Stichtag lässt die
Klassen täglich springen; der Abschluss ist nicht
reproduzierbar.

10 MONATSENDE und EDATUM
=EDATUM(Ausgangsdatum;Monate)
=MONATSENDE(Ausgangsdatum;Monate)
BEISPIEL B2 = 15.11.2026, C2 = 3 (Monate):
=MONATSENDE(EDATUM(B2;C2);0) → 28.02.2027
TYPISCHER FEHLER Ohne Datumsformat erscheint eine
Seriennummer (46446) statt eines Datums.

